

Investment Evaluation Program

FRED BLEECHMAN

HAVE YOU EVER MADE AN INVESTMENT IN stocks or metals and then wondered some time later if you would have been better off leaving the money in the bank? Or perhaps you'd like to know how much an investment must grow before it equals what you'd make leaving the money in the bank, at regular bank interest. The Investment Evaluation Program, written in TRS-80 Level I BASIC uses only 2464 bytes of RAM (Random Access Memory), so it can be run on the least expensive 4K RAM TRS-80.

The program is very straightforward, and has a handy subroutine for calculating the number of days between any two 20th century dates. The calculations are based on daily compounding of interest; if you want to change to monthly, quarterly or yearly compounding, you'll have to change lines 160, 170, 200, 250 and the subroutine starting at line 500.

Using the program is easy! After carefully entering and checking each line, type RUN and enter.

To illustrate, let's say that on July 20, 1974 you purchased 404 ounces of silver bullion for \$2098.38, including a service charge of 60¢ per ounce and 6% sales tax. You would like to know how the value of that bullion now compares with the same amount of money if it had been left in savings at, say, 5.25% annual interest, compounded daily. The day you want to calculate up to is July 1, 1978.

Set up the program on the computer and enter your name, social security number, 2098.38 invested at 5.25 as the questions are asked. Enter 0 for the days calculation, then enter 7,20,74 for the start date and 7,1,78 for the end date. The computer will tell you the number of investment days is 1442. Enter this number and the computer will print out, after about 40 seconds, the total interest (\$483.58) and the new principal of \$2581.80. (Don't mind the 16¢ error in addition of the interest and new principal. As a matter of fact, even the interest calculated is slightly off, due to rounding-off during the 1442 multiplications!)

Enter 404 for the number of ounces of silver bullion and the value of silver per ounce on the calculation end date—say, 5.28. The computer now displays that your investment is worth \$2133.12 and that you have now lost \$448.676 (there's that slight inaccuracy again!) compared to having left your money in your savings account. It also tells you that silver on that date would have to be worth \$6.39058 per ounce for you to just break even!

You can then press break to end the program, or perform another calculation. This same program can also be used to determine the future value of stock, bonds, gold, silver, etc., for break-even at some future date if you plan a particular investment. **R-E**

R-E will publish reader letters telling how to adopt this program to run on other hobby computers. Let us hear from you

```

100 CLS:P.
105 REM * COPYRIGHT 1978 FRED BLECHMAN * FOR TRS-80 I-4K *
110 P." INVESTMENT EVALUATION":P.:P.
112 P."THIS PROGRAM COMPARES AN INVESTMENT WITH PUTTING THE"
113 P."SAME AMOUNT OF MONEY IN A SAVINGS ACCOUNT WHERE IT EARNS"
114 P."DAILY INTEREST. IF YOU WISH TO CHANGE THE PERIOD TO"
115 P."MONTHLY OR YEARLY, CHANGE LINES 160,170,200 & 250. . ."
116 P." . . . . AND THE SUBROUTINE STARTING AT LINE 500."
120 P.:P.
125 IN."WHAT IS YOUR FIRST NAME":A$:
126 IN."WHAT IS YOUR SOCIAL SECURITY NUMBER":B$:
130 A=0:B=0:C=0:D=0:E=0:F=0:G=0:H=0:I=0
131 J=0:K=0:L=0:M=0:N=0:O=0:P=0:Q=0:R=0
132 S=0:T=0:U=0:V=0:W=0:X=0:Y=0:Z=0
140 IN."WHAT IS THE DOLLAR AMOUNT INVESTED":P
150 IN."WHAT IS YOUR REGULAR SAVINGS INTEREST RATE(%)":R
155 P.
160 P."HOW MANY DAYS ARE INVOLVED? IF YOU WANT THE NUMBER"
170 P."OF DAYS CALCULATED (20TH CENTURY ONLY) ENTER 0":D
180 IF D=0 GOSUB 500
185 IF D=0 GOTO 155
186 P.
190 P." . . . . .PATIENCE! . . . I'M CALCULATING THE ANSWER. . . ."
195 P." (TAKES ME ABOUT 10 SECONDS FOR 365 DAYS)"
200 S=R/36500:V=P
205 REM * CALCULATE INTEREST AND ADD TO PRINCIPAL *
210 FOR X=1 TO D
220 I=V*S:V=V+I:T=T+I
230 NEXT X
235 P.
240 P." THE TOTAL INTEREST IS":T
250 P."THE VALUE OF $":P;"AFTER":D;"DAYS AT":R;"% IS":V
255 P.
256 REM *COMPARE PRESENT VALUE OF INVESTMENT TO SAVINGS *
260 P."HOW MANY SHARES,BARS,OUNCES,ETC.,DO YOU OWN?"
265 IN." (TO RECALCULATE INVESTED AMOUNT,ENTER 0)":H
270 IF H=0 GOTO 130
275 P.
280 IN."WHAT IS THE PRESENT VALUE OF EACH SHARE,BAR,ETC.":M
290 Q=H*M
295 P.
300 P."YOUR INVESTMENT IS NOW WORTH $":Q;"":A$;"":
310 Z=V-Q
315 P.
320 IF Z>0 P.A$;"-":B$;"YOU HAVE LOST $":Z;"COMPARED TO SAVING!"
330 IF Z<0 P.A$;"-":B$;"YOU HAVE EARNED $":-Z;"MORE THAN SAVINGS!"
335 P.:P."THE 'BREAK-EVEN' POINT IS $":V/H;"SHARES,BARS,ETC."
340 P.:P."PRESS BREAK TO END PROGRAM. . . . ."
350 GOTO 130
360 END
500 REM * SUBROUTINE FOR CALCULATING DAYS *
510 DATA 0,31,28,31,30,31,30,31,31,30,31,30,31
520 REM * DETERMINE NUMBER OF DAYS FROM 0 TO START *
525 P.
530 IN."WHAT IS THE INVESTMENT START DATE(M,D,Y)":A,B,C
540 E=A
550 GOSUB 1000
560 F=F+B
570 G=F+C*365
580 REM * DETERMINE NUMBER OF DAYS FROM 0 TO END *
590 IN."WHAT IS THE INVESTMENT END DATE(M,D,Y)":J,K,L
600 E=J
610 GOSUB 1000
620 F=F+K
630 N=F+L*365
640 REM * CALCULATE AND ADD LEAP YEARS *
650 O=INT((L-1900)/4):U=INT((C-1900)/4):W=O-U
660 X=N-(G)+W
665 P.
670 P."THE NUMBER OF INVESTMENT DAYS IS":X
680 RETURN
1000 F=0
1010 FOR X=1 TO E
1020 READ Y
1030 IF Y=28 THEN IF (L/4)-INT(L/4)=0 THEN Y=29
1040 F=F+Y
1050 NEXT X
1060 RESTORE
1070 RETURN

```